Computer Games

karamellpelle@hotmail.com

March 29, 2012

Abstract

The reason I got interested in programming is because I wanted to create computer games. I tried to find out how computer game programs should be organized, but I couldn't find any information about this. So I had to organize the game programs my own way, and this is the (current) result. I would be glad if somebody can learn me more game programming.

Here we make an attempt to define a computer game and construct such a thing in Haskell, with focus on implementing the game flow of the program.

Definition

A computer game is the side effects of the following system:



At each frame t there is an iteration i_t which iterate the current world. This gives sequences

$$\mathtt{i}_0 \longrightarrow \mathtt{i}_1 \longrightarrow \mathtt{i}_2 \longrightarrow \ldots$$

$$\mathtt{w}_0 \longrightarrow \mathtt{w}_1 \longrightarrow \mathtt{w}_2 \longrightarrow \ldots$$

$$\mathtt{o}_0 \longrightarrow \mathtt{o}_1 \longrightarrow \mathtt{o}_2 \longrightarrow \ldots$$

where $\mathtt{i}_t$ is the input state, $\mathtt{w}_t$ is the game state, $\mathtt{o}_t$ is the output state, at frame t.

Let's construct such a system in Haskell.

The environment

The environment usually consist of input, output, time, random and resources used by the game (time, random and resources are not considered as input/output). The input will typically be

- Keys
  - Keyboard
  - Mouse
  - Joystick
  - etc.
- Network

The output will typically be

- Screen
- Network
- Sound

The world

The world data is game specific. But usually the world data will contain the following fields:

```
data World =
    World
    {
        worldTime :: Time,
        worldEvents :: [WorldEvent],
        -- ...
    }
```

where `WorldEvent` is game specific.

Computer games inside computer games

It might be reasonable to consider a computer game as computer games inside computer games. For example, a game may consist of a set of levels, making a story. Hence we can think of `LevelWorld` as contained inside `StoryWorld`. Then we can let an iteration of `StoryWorld` iterate `LevelWorld` in order to play that level of the story. `StoryWorld` may again be contained in `RunWorld` in order to run the whole game.

Construction

Let us model computations inside an environment as a monad `MEnv`. If `a` is a world contained inside a world `b`, we let an iteration of `a` work on both `a` and `b`. `a` is the main world the iteration is working on. We will use a stack of iterations, and at each frame we pop the top element of the current stack (if any) and perform that iteration on the worlds `a` and `b`, modifying the worlds and giving a new top on the stack.

```
data Iteration a b =
    Iteration
    {
```

```
        iteration :: a -> b -> MEnv (a, b, IterationStack a b)
    }


type IterationStack a b =
    [Iteration a b]


-- | running a stack on 'a' and 'b'
runABStack :: a -> b -> IterationStack a b -> MEnv (a, b)
runABStack a b stack =
    case stack of
        []         -> return (a, b)
        (i:is)     -> do

            -- (begin MEnv frame)

            (a', b', top) <- (iteration i) a b

            -- (end MEnv frame)

            runABStack a' b' (top ++ is)


-- | iterating a stack on 'a' and 'b'
--    (typically to be done inside another iteration)
iterateABStack :: a -> b -> IterationStack a b ->
                  MEnv (a, b, IterationStack a b)
iterateABStack a b stack =
    case stack of
        []         -> return (a, b, [])
        (i:is)     -> do
            (a', b', top) <- (iteration i) a b
            return (a', b', top ++ is)
```

Iteration

In our construction we want to, at each frame, output the current world and then step it. Hence we create a function `defaultIteration`:

```
defaultIteration ::
                s ->
```

```
                  (s -> a -> b -> MEnv s) ->
                  (s -> a -> b -> MEnv (a, b, IterationStack a b)) ->
                  Iteration a b
defaultIteration s output step =
    Iteration $ \a b -> do
        s' <- output s a b
        step s' a b
```

The parameter `s` makes it possible for an iteration to work with a state.

Output

The output is game and environment specific. The typical things to do are screen drawing, send data over network and playing sounds.

Step

The task of the step part is to modify the worlds and modify the stack of iterations. Since we have output at each frame, `defaultStep` is “do–think” instead of “think–do”.

```
defaultStep :: (s -> a -> b -> MEnv (s, a, b)) ->
               (s -> a -> b -> MEnv (a, b, IterationStack a b)) ->
               s -> a -> b -> MEnv (a, b, IterationStack a b)
defaultStep doWorld
            thinkWorld = \s a b -> do

    -- do
    (s', a', b') <- doWorld s a b

    -- think
    thinkWorld s' a' b'
```

Do

A world may consist of physical objects which are modified by time, and thus we define a function `defaultDo`:

```
defaultDo :: (s -> a -> b -> MEnv (s, a, b)) ->
             (MEnv Time, Time -> MEnv (), Time, Time) ->
```

```
             (Time -> s -> a -> b -> MEnv (s, a, b)) ->
             (s -> a -> b -> MEnv (Maybe (s, a, b))) ->
             (s -> a -> b -> MEnv (s, a, b)) ->
             s -> a -> b -> MEnv (s, a, b)
defaultDo modify
     (getTime, setTime, dtUnit, maxElaps)
     stepDT
     breakModify
     defaultModify = \s a b -> do

        -- begin modify world
        (s', a', b') <- modify s a b

        -- ignore too long elaps; prevent program hang
        time <- getTime
        when ( worldTime a' + maxElaps <= time ) $
            setTime ( worldTime a' + maxElaps )
        time <- getTime

        -- step physics in 'dtUnit' portions
        helper time s' a' b'
        where
          helper time s a b =
            if worldTime a + dtUnit <= time

              -- take a 'dtUnit'-step of physical objects
              then do
                (s', a', b') <- stepDT dtUnit s a b
                maybeSAB <- breakModify s' a' b'
                case maybeSAB of
                    -- continue
                    Nothing              -> helper time s' a' b'
                    -- end modify world
                    Just (s'', a'', b'')  -> return (s'', a'', b'')

              -- end modify world
              else do
                defaultModify s a b
```

After a first modification of the world, we step the physical part of the world by elapsed time. This is done in constant `dtUnit` timesteps. Each `stepDT` increments the world's `worldTime` with the value of `dtUnit`. After each `stepDT` we may escape from further steps and then modify the world with `breakModify`, or continue until no more dt–steps can be done (because the difference between time and worldTime is less than dtUnit; worldTime is very up to date), whereat we modify the worlds with defaultModify.

Modify

A first modification of the world before the physical step (stepDT) may look like:

```
defaultModify :: (s -> a -> b -> MEnv (s, a, b)) ->
                 (s -> a -> b -> MEnv (s, a, b)) ->
                 (s -> a -> b -> MEnv (s, a, b)) ->
                 s -> a -> b -> MEnv (s, a, b)

defaultModify beginModify
              controlModify
              updateModify = \s a b -> do

    -- modify world at beginning of step (like clean up)
    (s', a', b') <- beginModify s a b

    -- modify world from controls (like keys, network, AI, ...)
    (s'', a'', b'') <- controlModify s' a' b'

    -- modify world by updating it (like timers, ...)
    (s''', a''', b''') <- updateModify s'' a'' b''

    return (s''', a''', b''')
```

StepDT

The stepDT function is world and game specific. The function typically also checks for collisions between physical objects, so an implementation may be like:

```
stepDT :: CollisionHandler ->
          Time ->
          s -> a -> b -> MEnv (s, a, b)
stepDT handleCollision dt = \s a b -> do
    -- make a 'dt' step of physical objects. check for collisions
    -- between such, and use 'handleCollision' to make changes to
    -- world. example of such changes to world is pushing
    -- WorldEvent's to world. increment worldTime by 'dt'.
    -- ...
```

Think

At iteration end, we modify the worlds and decide the next top of the iteration stack. An iteration `A` might decide stack tops of different sorts:

`[]`	finish iteration `A`
`[B]`	finish iteration `A`, new iteration `B`
`[B, C]`	finish iteration `A`, new iteration `B`, iteration `C` on finish
`[A]`	continue iteration `A`
`[B, A]`	new iteration `B`, continue with iteration `A` on finish
`[A, B]`	continue iteration `A`, iteration `B` on finish

(this is not a complete characterization). So it is possible to construct a sequence of iterations `A` → `B` → `C` differently: iteration `A` returns `[B, C]`, and iteration `A` returns `[B]`, iteration `B` returns `[C]`. In the second case, iteration `B` needs to be aware of iteration `C`, but in the first case, it does not. Hence the second construction is more specialized than the first construction.

Let's make an attempt to draw the flow of iterations:



This should be read as:

- In addition to `[]` and `[A]`, iteration `A` may return either `[B, C, D]` or `[E]`.
- In addition to `[]` and `[B]`, iteration `B` may return `[F]`.
- Iteration `C` returns `[]` or `[C]`.
- In addition to `[]` and `[D]`, iteration `D` may return `[G, D]`.
- Iteration `E` returns `[]` or `[E]`.
- Iteration `F` returns `[]` or `[F]`.
- Iteration `G` returns `[]` or `[G]`

Hence we read a box with an arrow as "on finish".

Iteration modifiers

We can make utility functions to manipulate iterations, changing how they behave, like:

```
-- | modify worlds before iteration
modifyBefore :: Iteration a b ->
                (a -> b -> MEnv (a, b)) ->
                Iteration a b

-- | modify worlds after iteration
modifyAfter :: Iteration a b ->
               (a -> b -> MEnv (a, b)) ->
               Iteration a b

-- | run iteration with local world, then continue with original
--   world when finished
localWorldA :: a ->
               Iteration a b ->
               Iteration a b

-- | run iteration, then continue with original world when finished
saveWorldA :: Iteration a b ->
              Iteration a b

-- | when finished, choose next iterations by looking at worlds
chooseStackAfter :: Iteration a b ->
                    (a -> b -> MEnv (IterationStack a b)) ->
                    Iteration a b
```

They can in theory be used for special effects in game as jumping back in time with `localWorldA`, jumping forward in time with `saveWorldA`, etc.